


# OBSTACLE



## CLASSIFICATION: Avoidance Game

Use 'A' and 'Z' to move (up and down respectively) through the obstacle course.

Your goal is to reach the bottom in the best possible time, avoiding the stompers and lasers along the way. At the lower set of stompers, you only need to move one step below the lasers and then you will fall all the way.

## PROGRAMMING SUGGESTIONS

To make the game more difficult the number of lasers or the number of stompers could be increased. To make it impossible increase both.

## **PROGRAM Variables**

TM	Time
PHASE	Phase man has reached (1-3)
MROW	Man's row
LASER(10)	Timer counts for each laser
LASROW(10)	Row of each laser
ASTOMPLXL(5),	Top set stompers
ASTOMPXR(5)	columns (left and right)
BSTOMPXL(3),	Bottom set
BSTOMPXR(3)	stompers columns (left and right)
ASTOMPY(5)	Top stompers rows
BSTOMPY(3)	Bottom stompers rows
ASTOMPDR(5)	Top stompers directions
BSTOMPDR(3)	Bottom stompers directions
I, J, K	Temps

## **Program Structure**

Lines	Function/Activity
10 — 110	Initialise
130 — 500	Program main loop
1000 — 1060	Fire lasers
2000 — 2630	Move stomper
8000	Update time
9000 — 9020	Game over

OBSTACLE

Initialise game

```

10  MODE 1 : INK 0, 0 : INK 1, 6 : INK 2, 11 : INK 3, 24
20  FOR I = 1 TO 10 : LASROW( I ) = 8 + I : NEXT
30  PHASE = 1 : MROW = 2
40  FOR I = 1 TO 5 : ASTOMPY( I ) = 3 + I : ASTOMPXL( I ) =
    15 : ASTOMPXR( I ) = 24 : ASTOMPDR( I ) = 1 : NEXT
50  FOR I = 1 TO 3 : BSTOMPY( I ) = 17 + 2 * I : BSTOMPXL(
    I ) = 15 : BSTOMPXR( I ) = 24 : BSTOMPDR( I ) = 1 : NEXT

60  PAPER 0 : CLS : BORDER 0
70  EVERY 50 GOSUB 8000
80  PEN 2 : FOR I = 1 TO 10 : LOCATE 1, LASROW( I ) : PRINT
    CHR$( 246 ) : LOCATE 40, LASROW( I ) : PRINT CHR$( 247
    ) : NEXT
90  FOR I = 1 TO 5 : FOR J = 1 TO 14 : LOCATE J, ASTOMPY( I
    ) : PRINT CHR$( 154 ) : LOCATE J + 24, ASTOMPY( I ) : P
    RINT CHR$( 154 ) : NEXT : NEXT
100  FOR I = 1 TO 3 : FOR J = 1 TO 14 : LOCATE J, BSTOMPY( I
    ) : PRINT CHR$( 154 ) : LOCATE J + 24, BSTOMPY( I ) : P
    RINT CHR$( 154 ) : NEXT : NEXT
110  PEN 1 : LOCATE 20, MROW : PRINT CHR$( 249 )

```

Main Program loop

```

120  :
130  IF INKEY( 69 ) = -1 OR MROW < 3 THEN 140 ELSE LOCATE 20
    , MROW : PRINT "▲" : MROW = MROW - 1 : IF TEST( 304, 407
    - 16 * MROW ) <> 0 THEN 9000 ELSE PEN 1 : LOCATE 20, MR
    OW : PRINT CHR$( 249 )
140  IF PHASE = 3 THEN 145 ELSE IF INKEY( 71 ) = -1 OR MROW
    > 18 THEN 150
145  LOCATE 20, MROW : PRINT "▲" : MROW = MROW + 1 : IF TEST
    ( 304, 407 - 16 * MROW ) <> 0 THEN 9000 ELSE PEN 1 : LOC
    ATE 20, MROW : PRINT CHR$( 249 ) : IF MROW > 24 THEN 900
    0
150  PHASE = -( MROW > 1 ) - ( MROW > 8 ) - ( MROW > 18 )
160  PEN 1 : LOCATE 1, 1 : PRINT "TIME : ▲": TM
170  IF PHASE <> 1 THEN 190 ELSE I = INT( RND( 1 ) * 5 ) + 1
    : GOSUB 2000
180  FOR K = 1 TO 3 : I = INT( RND( 1 ) * 5 ) + 1 : GOSUB 20
    00 : NEXT
190  IF PHASE = 2 THEN GOSUB 1000
200  IF PHASE > 1 THEN FOR K = 1 TO 5 : I = INT( RND( 1 ) *
    3 ) + 1 : GOSUB 2000 : NEXT
500  GOTO 130

```

Fire lasers

```

990  :
1000  FOR I = 1 TO 10 : J = INT( RND( 1 ) * 10 ) + 5
1005  IF LASER( I ) < 0 THEN LASER( I ) = 0 : MOVE 17, 407 -
    16 * LASROW( I ) : DRAWR 605, 0, 0

```

```

1010 IF LASER( I ) = 0 THEN IF RND( 1 ) < 0.9 THEN 1060 ELSE
      LASER( I ) = J : PEN 3 : LOCATE 1, LASROW( I ) : PRINT
      CHR$( 246 ) : LOCATE 40, LASROW( I ) : PRINT CHR$( 247 )
      : GOTO 1060
1020 LASER( I ) = LASER( I ) - 1 : IF LASER( I ) > 0 THEN 10
      60
1030 PEN 2 : LOCATE 1, LASROW( I ) : PRINT CHR$( 246 ) : LOC
      ATE 40, LASROW( I ) : PRINT CHR$( 247 ) : MOVE 17, 407 -
      16 * LASROW( I )
1040 SOUND 1, 400, 10, 15 : DRAWR 605, 0, 3
1050 LASER( I ) = -1 : IF LASROW( I ) = MROW THEN 9000
1060 NEXT : RETURN

```

Move stomper

```

1990 :
2000 IF PHASE = 1 THEN 2500
2010 PEN 2 : LOCATE BSTOMPXL( I ), BSTOMPY( I ) : IF BSTOMPD
      R( I ) = 1 THEN PRINT CHR$( 154 ) ELSE PRINT "▲"
2020 LOCATE BSTOMPXR( I ), BSTOMPY( I ) : IF BSTOMPDR( I ) =
      1 THEN PRINT CHR$( 154 ) ELSE PRINT "▲"
2030 IF BSTOMPDR( I ) = -1 THEN 2100 ELSE IF BSTOMPXL( I ) =
      19 THEN IF BSTOMPY( I ) = MROW THEN 9000 ELSE BSTOMPDR(
      I ) = -1 : GOTO 2050
2040 BSTOMPXL( I ) = BSTOMPXL( I ) + BSTOMPDR( I ) : BSTOMPX
      R( I ) = BSTOMPXR( I ) - BSTOMPDR( I )
2050 LOCATE BSTOMPXL( I ), BSTOMPY( I ) : PRINT CHR$( 143 )
      : LOCATE BSTOMPXR( I ), BSTOMPY( I ) : PRINT CHR$( 143 )

2060 RETURN
2100 IF BSTOMPXL( I ) = 15 THEN BSTOMPDR( I ) = 1 : GOTO 212
      0
2110 BSTOMPXL( I ) = BSTOMPXL( I ) + BSTOMPDR( I ) : BSTOMPX
      R( I ) = BSTOMPXR( I ) - BSTOMPDR( I )
2120 LOCATE BSTOMPXL( I ), BSTOMPY( I ) : PRINT CHR$( 143 )
      : LOCATE BSTOMPXR( I ), BSTOMPY( I ) : PRINT CHR$( 143 )

2130 RETURN
2500 PEN 2 : LOCATE ASTOMPXL( I ), ASTOMPY( I ) : IF ASTOMPD
      R( I ) = 1 THEN PRINT CHR$( 154 ) ELSE PRINT "▲"
2510 LOCATE ASTOMPXR( I ), ASTOMPY( I ) : IF ASTOMPDR( I ) =
      1 THEN PRINT CHR$( 154 ) ELSE PRINT "▲"
2520 IF ASTOMPDR( I ) = -1 THEN 2600 ELSE IF ASTOMPXL( I ) =
      19 THEN IF ASTOMPY( I ) = MROW THEN 9000 ELSE ASTOMPDR(
      I ) = -1 : GOTO 2540
2530 ASTOMPXL( I ) = ASTOMPXL( I ) + ASTOMPDR( I ) : ASTOMPX
      R( I ) = ASTOMPXR( I ) - ASTOMPDR( I )
2540 LOCATE ASTOMPXL( I ), ASTOMPY( I ) : PRINT CHR$( 143 )
      : LOCATE ASTOMPXR( I ), ASTOMPY( I ) : PRINT CHR$( 143 )

2550 RETURN
2600 IF ASTOMPXL( I ) = 15 THEN ASTOMPDR( I ) = 1 : GOTO 262
      0
2610 ASTOMPXL( I ) = ASTOMPXL( I ) + ASTOMPDR( I ) : ASTOMPX
      R( I ) = ASTOMPXR( I ) - ASTOMPDR( I )
2620 LOCATE ASTOMPXL( I ), ASTOMPY( I ) : PRINT CHR$( 143 )
      : LOCATE ASTOMPXR( I ), ASTOMPY( I ) : PRINT CHR$( 143 )

2630 RETURN

```

Update time

```
7990 :  
8000 TM = TM + 1 : RETURN
```

Game over

```
8990 :  
9000 LOCATE 15, 1 : PRINT "GAME_OVER-SCORE="; MROW * 10  
- TM * 2 - 100 * ( MROW > 23 )  
9005 FOR I = 1 TO 1500 : NEXT  
9007 IF MROW > 23 THEN LOCATE 1, 2 : PEN 3 : PRINT "CONGRATU  
LATIONS!! YOU HAVE WON!!"  
9010 IF INKEY$ = "" THEN 9020 ELSE 9010  
9020 IF INKEY$ = "" THEN 9020 ELSE RUN
```

## ChexSum Tables

10 = 1357	990 = 0	2510 = 5393
20 = 2798	1000 = 2760	2520 = 8684
30 = 1486	1005 = 4952	2530 = 7392
40 = 6882	1010 = 9352	2540 = 6113
50 = 7171	1020 = 3747	2550 = 201
60 = 548	1030 = 6203	2600 = 3472
70 = 612	1040 = 1051	2610 = 7392
80 = 5463	1050 = 3521	2620 = 6113
90 = 7015	1060 = 378	2630 = 201
100 = 7047	1990 = 0	7990 = 0
110 = 1860	2000 = 1491	8000 = 1304
120 = 0	2010 = 5594	8990 = 0
130 = 9990	2020 = 5396	9000 = 4932
140 = 4293	2030 = 8431	9005 = 1386
145 = 8515	2040 = 7366	9007 = 4292
150 = 3951	2050 = 6085	9010 = 1545
160 = 1588	2060 = 201	9020 = 1632
170 = 3845	2100 = 3452	
180 = 3337	2110 = 7398	
190 = 1713	2120 = 6117	
200 = 4624	2130 = 201	TOTAL = 233818
500 = 352	2500 = 5623	